

# VAE – AE with Variables

## 1 INTRODUCTION

VAE is a toy language for the [COSE212](#) course at Korea University. VAE stands for an extension of the [AE](#) language with **variables**, and it supports the following features:

- **integers**
- **basic arithmetic operators**: addition (+) and multiplication (\*)
- **immutable variables** (val)

This document is the specification of VAE. First, Section 2 describes the concrete syntax, and Section 3 describes the abstract syntax. Then, Section 4 describes the big-step operational (natural) semantics of VAE.

## 2 CONCRETE SYNTAX

The concrete syntax of VAE is written in a variant of the extended Backus–Naur form (EBNF). The notation `<nt>` denotes a nonterminal, and `"t"` denotes a terminal. We use `?` to denote an optional element and `+` (or `*`) to denote one or more (or zero or more) repetitions of the preceding element. We use **butnot** to denote a set difference to exclude some strings from a producible set of strings. We omit some obvious terminals using the ellipsis (`...`) notation.

```
// basic elements
<digit>    ::= "0" | "1" | "2" | ... | "9"
<number>   ::= "-"? <digit>+
<alphabet> ::= "A" | "B" | "C" | ... | "Z" | "a" | "b" | "c" | ... | "z"
<idstart>  ::= <alphabet> | "_"
<idcont>   ::= <alphabet> | "_" | <digit>
<keyword>  ::= "val"
<id>       ::= <idstart> <idcont>* butnot <keyword>

// expressions
<expr>     ::= <number> | <expr> "+" <expr> | <expr> "*" <expr>
              | "(" <expr> ")" | "{" <expr> "}"
              | "val" <id> "=" <expr> ";" <expr> | <id>
```

The precedence and associativity of operators are defined as follows:

Operator	Associativity	Precedence
*	left	1
+	left	2

## 3 ABSTRACT SYNTAX

The abstract syntax of VAE is defined as follows:

Expressions	$e ::= n$	(Num)	Environments	$\sigma \in \mathbb{X} \xrightarrow{\text{fin}} \mathbb{Z}$	(Env)
	$  e + e$	(Add)		$n \in \mathbb{Z}$	(BigInt)
	$  e \times e$	(Mul)		$x \in \mathbb{X}$	(String)
	$  \text{val } x = e; e$	(Val)			
	$  x$	(Id)			

#### 4 SEMANTICS

The big-step operational (natural) semantics of VAE is defined as follows:

$$\boxed{\sigma \vdash e \Rightarrow n}$$

$$\begin{array}{c} \text{Num} \frac{}{\sigma \vdash n \Rightarrow n} \quad \text{Add} \frac{\sigma \vdash e_1 \Rightarrow n_1 \quad \sigma \vdash e_2 \Rightarrow n_2}{\sigma \vdash e_1 + e_2 \Rightarrow n_1 + n_2} \quad \text{Mul} \frac{\sigma \vdash e_1 \Rightarrow n_1 \quad \sigma \vdash e_2 \Rightarrow n_2}{\sigma \vdash e_1 \times e_2 \Rightarrow n_1 \times n_2} \\ \\ \text{Val} \frac{\sigma \vdash e_1 \Rightarrow n_1 \quad \sigma[x \mapsto n_1] \vdash e_2 \Rightarrow n_2}{\sigma \vdash \text{val } x = e_1; e_2 \Rightarrow n_2} \quad \text{Id} \frac{x \in \text{Domain}(\sigma)}{\sigma \vdash x \Rightarrow \sigma(x)} \end{array}$$