VAE – Arithmetic Expressions with Variables

1 INTRODUCTION

VAE is a toy language for the COSE212 course at Korea University. VAE stands for a language of **a**rithmetic **e**xpressions with **v**ariables, and it supports the following features:

- **integers**
- **basic arithmetic operators**: addition (`+`) and multiplication (`*`)
- **immutable variables** (`val`)

This document is the specification of VAE. First, Section 2 describes the concrete syntax, and Section 3 describes the abstract syntax. Then, Section 4 describes the big-step operational (natural) semantics of VAE.

2 CONCRETE SYNTAX

The concrete syntax of VAE is written in a variant of the extended Backus–Naur form (EBNF). The notation `<nt>` denotes a nonterminal, and `"t"` denotes a terminal. We use `?` to denote an optional element and `+` (or `*`) to denote one or more (or zero or more) repetitions of the preceding element. We omit some obvious terminals using the ellipsis (`...`) notation.

```
<digit>   ::= "0" | "1" | "2" | ... | "9"
<number>  ::= "-"? <digit>+
<letter>  ::= "A" | "B" | "C" | ... | "Z" | "a" | "b" | "c" | ... | "z"
<idstart> ::= <letter> | "_"
<idcont>  ::= <idstart> | <digit>
<id>      ::= <idstart> <idcont>*
<expr>    ::= <number> | <expr> "+" <expr> | <expr> "*" <expr>
            | "(" <expr> ")" | "{" <expr> "}"
            | "val" <id> "=" <expr> ";" <expr> | <id>
```

The precedence and associativity of operators are defined as follows:

Operator	Associativity	Precedence
*	left	1
+	left	2

3 ABSTRACT SYNTAX

The abstract syntax of VAE is defined as follows:

$$
\begin{array}{lll}
\text{Expressions} & e ::= n & \texttt{(Num)} \\
& \mid e + e & \texttt{(Add)} \\
& \mid e \times e & \texttt{(Mul)} \\
& \mid \texttt{val}\ x = e;\ e & \texttt{(Val)} \\
& \mid x & \texttt{(Id)}
\end{array}
$$

$$
\begin{array}{lll}
\text{Environments} & \sigma \in \mathbb{X} \xrightarrow{\text{fin}} \mathbb{Z} & \texttt{(Env)} \\
\text{Integers} & n \in \mathbb{Z} & \texttt{(BigInt)} \\
\text{Identifiers} & x \in \mathbb{X} & \texttt{(String)}
\end{array}
$$

4 SEMANTICS

The big-step operational (natural) semantics of VAE is defined as follows:

$$\boxed{\sigma \vdash e \Rightarrow n}$$

$$\texttt{Num}\ \frac{}{\sigma \vdash n \Rightarrow n} \qquad \texttt{Add}\ \frac{\sigma \vdash e_1 \Rightarrow n_1 \qquad \sigma \vdash e_2 \Rightarrow n_2}{\sigma \vdash e_1 + e_2 \Rightarrow n_1 + n_2} \qquad \texttt{Mul}\ \frac{\sigma \vdash e_1 \Rightarrow n_1 \qquad \sigma \vdash e_2 \Rightarrow n_2}{\sigma \vdash e_1 \times e_2 \Rightarrow n_1 \times n_2}$$

$$\texttt{Val}\ \frac{\sigma \vdash e_1 \Rightarrow n_1 \qquad \sigma[x \mapsto n_1] \vdash e_2 \Rightarrow n_2}{\sigma \vdash \texttt{val}\ x = e_1;\ e_2 \Rightarrow n_2} \qquad \texttt{Id}\ \frac{x \in \mathrm{Domain}(\sigma)}{\sigma \vdash x \Rightarrow \sigma(x)}$$